数学实验第七次实验报告

计 76　张翔　2017011568

2020 年 4 月 24 日

1　实验目的

1. 练习建立实际问题的整数规划模型;
2. 掌握用 LINGO 软件求解整数规划问题。

2　Ch10-P8 服务员聘用

2.1　问题分析与模型建立

对于储蓄所来说，应在满足每个时段人数要求的情况下，尽可能降低雇佣成本。为了建立合适的数学模型，可以对题目做如下简化

- 储蓄所一定能雇佣到给定数量的员工，不会出现供不应求；
- 所有服务员能在规定时间内正常工作，不会出现请假等情况；
- 工作起始时间、午餐起始时间均在整点。

根据上述假设, 可以将全时服务员划分为 2 个等价类, 其午餐起始时间分别为 12am, 2pm, 用 x_1, x_2 代表两类各自的人数; 半时服务员划分为 5 个等价类, 其工作起始时间分别为 9am, 10am, 11am, 12am, 1pm，将每一类的人数记作 y_1, y_2, y_3, y_4, y_5。由此，可以建立下面的规划模型：

$$\min z = 100(x_1 + x_2) + 40(y_1 + y_2 + y_3 + y_4 + y_5)$$

需要满足每个时间段的人数约束

$$\begin{aligned}
x_1 + x_2 + y_1 &\geq 4 \\
x_1 + x_2 + y_1 + y_2 &\geq 3 \\
x_1 + x_2 + y_1 + y_2 + y_3 &\geq 4 \\
x_2 + y_1 + y_2 + y_3 + y_4 &\geq 6 \\
x_1 + y_2 + y_3 + y_4 + y_5 &\geq 5 \\
x_1 + x_2 + y_3 + y_4 + y_5 &\geq 6 \\
x_1 + x_2 + y_4 + y_5 &\geq 8 \\
x_1 + x_2 + y_5 &\geq 8
\end{aligned}$$

上述规划变量均隐含了非负整数的约束。

对于问题 (1)，增加每天雇佣不超过 3 名半时服务员的限制

$$y_1 + y_2 + y_3 + y_4 + y_5 \leq 3$$

对于问题 (2)，不能雇佣半时服务员，则要求

$$y_1 + y_2 + y_3 + y_4 + y_5 \leq 0$$

对于问题 (3)，没有雇佣半时服务员的限制，无需增加约束。

2.2 算法设计

上述目标函数、约束条件均为线性，且规划变量为非负整数，是整数线性规划问题，可以使用 LINGO 自带的整数规划功能进行求解。

2.3 LINGO 代码

第 (1) 问代码如下

```
MODEL:
TITLE: Employment;
[OBJ] MIN = 100*(x1+x2)+40*(y1+y2+y3+y4+y5);
x1+x2+y1 >= 4;
x1+x2+y1+y2 >= 3;
x1+x2+y1+y2+y3 >= 4;
x2+y1+y2+y3+y4 >= 6;
x1+y2+y3+y4+y5 >= 5;
x1+x2+y3+y4+y5 >= 6;
x1+x2+y4+y5 >= 8;
x1+x2+y5 >= 8;
[P1]y1+y2+y3+y4+y5 <= 3;
@gin(x1);@gin(x2);
@gin(y1);@gin(y2);@gin(y3);@gin(y4);@gin(y5);
END
```

第 (2) 问只需要修改上述 [P1] 约束即可

```
[P2]y1+y2+y3+y4+y5 <= 0;
```

第 (3) 问只需要去掉上述 [P1] 约束。

2.4 计算结果与分析

(1) 使用题给数据，同时雇佣半时、全时服务员，且半时服务员不超过 3 人时，结果如下

$$x_1 = 2,\ x_2 = 5,\ y_2 = 1,\ y_5 = 2,$$

$$y_1 = y_3 = y_4 = 0$$

目标函数最优值为 $z = 820$，其中 12am, 1pm 的人数约束、半时服务员总人数约束的松弛变量为 0，目标函数最优时它们取等号。

(2) 如果不能雇佣半时服务员，可以得到下列结果

$$x_1 = 5,\ x_2 = 6,\ y_1 = y_2 = y_3 = y_4 = y_5 = 0$$

目标函数最优值为 $z = 1100$，其中 12am, 1pm 的人数约束松弛变量为 0，目标函数最优时它们取等号。

(3) 如果不限制雇佣半时服务员，则得到结果

$$x_1 = x_2 = y_2 = y_3 = 0, y_1 = 4, y_4 = 2, y_5 = 8$$

目标函数最优值为 $z = 560$，其中 9am, 11am, 12am, 4pm 的人数约束松弛变量为 0，目标函数最优时它们取等号。

比较上述数据，可以发现，约束半时服务员总人数为 3 时，总雇佣成本为 820 元；如果不能雇佣半时服务员，总雇佣成本大幅上升，为 1100 元，每天增加了 280 元；如果不限制雇佣半时服务员，总成本只要 560 元，每天减少 540 元。可见是否雇佣半时员工、雇佣半时的人数对最终成本起到了重要作用。

分析题目条件，可看出半时服务员的时薪为 10 元/小时，而全时服务员虽然 9am 到岗，5pm 下班，但中途有 1 小时的午餐时间，实际工作时间只有 7 小时，平均时薪 14.29 元/小时。因此，从平均时薪的角度来说，雇佣半时服务员相比全时服务员是绝对划算的。

另外，从时段需求人数分析，12am-1pm, 1pm-2pm 这两段时间内需要的人数在中等水平（接近一天内需求的中位数 5.5），而此时恰好是全时工的午餐时间，这进一步削弱了雇佣全时工的成本优势。在模型的假设中，雇佣 2 个半时员工，上班时间分别为 9am，1pm，等效于 1 个不需要午餐时间的全时员工，且雇佣费用更少，更加划算。因此，情况 3 中不限制雇佣人数时，自然不需要雇佣全时工作人员。

2.5 结论

1. 雇佣半时服务员不超过 3 人时，最小雇佣成本 820 元，此时雇佣 2 个在 12am-1am 午餐、5 个 1am-2am 午餐的全时服务员，1 个 10am 开始工作、2 个 1am 开始工作的半时服务员；
2. 只能雇佣全时服务员时，最小雇佣成本较第一种情况增加 280 元，此时雇佣 5 个在 12am-1am 午餐、6 个 1am-2am 午餐的全时服务员；
3. 不限制雇佣人数时，最小雇佣佣成本较第一种情况减少 540 元，此时不雇佣全时工作人员，雇佣 4 个 9am 上班、2 个 12am 上班、8 个 1am 上班的半时工作人员；
4. 条件允许的情况下，应尽量只招半时服务员，从而更好地降低成本。

3 Ch10-P9 原油采购与加工

3.1 问题分析与模型建立

对于题中的公司来说，需要设计合适的原油购买量，以得到最大的净利润。由于净利润为总利润 − 成本，使用原油时一定优先使用存货（成本为 0），存货不够时才考虑加以购买。

设最后成品油中，甲含 A,B 原油分别为 a_1, b_1 吨，乙含 A,B 原油分别为 a_2, b_2 吨。库存使用量 A,B 原油分别为 x, y 吨。由于原油 A 的市场价格是分段函数，不妨假设三段价格的购买量分别为 t_1, t_2, t_3 吨，那么模型建立如下

目标函数

$$\max z = 4800(a_1 + b_1) + 5600(a_2 + b_2) - 10000t_1 - 8000t_2 - 6000t_3$$

约束条件如下

油量守恒

$$a_1 + a_2 = x + t_1 + t_2 + t_3$$

$$b_1 + b_2 = y$$

成品的含油比例

$$a_1 \geq 0.5(a_1 + b_1)$$

$$a_2 \geq 0.6(a_2 + b_2)$$

库存量限制

$$x \leq 500,\ y \leq 1000$$

购买原油 A 的分段约束

$$t_1 \leq 500,\ t_2 \leq 500,\ t_3 \leq 500$$

$$t_2(t_1 - 500) = 0$$

$$t_3(t_2 - 500) = 0$$

所有决策变量均隐含了**非负**这一约束条件。

3.2 算法设计

题目要求使用连续规划和整数规划模型进行求解，因此可以使用 LINGO 在连续规划的基础上添加 `@gin` 以使用整数规划模型。该问题出现了非线性约束条件，是非线性的规划问题。

3.3 LINGO 代码

连续规划的代码如下

```
MODEL:
TITLE Oil purchase;
[OBJ] max = 4800 * (a1 + b1) + 5600 * (a2 + b2) - 10000 * t1 - 8000 * t2 - 6000 * t3;
a1 + a2 = x + t1 + t2 + t3;
b1 + b2 = y;
a1 >= 0.5 * (a1 + b1);
a2 >= 0.6 * (a2 + b2);
x <= 500;
y <= 1000;
t1 <= 500;
t2 <= 500;
t3 <= 500;
t2 * (t1 - 500) = 0;
t3 * (t2 - 500) = 0;
```

整数规划只需要将上述决策变量添加 `@gin` 即可，在此不再赘述。

3.4 计算结果与分析

连续规划 LINGO 使用二次规划 QP 模型进行求解，求解器迭代次数 1374，运行时间 0.13s，求解结果如下

$$a_1 = b_1 = 0,\ a_2 = 1500,\ b_2 = 1000,$$

$$t_1 = 500,\ t_2 = 500,\ t_3 = 0,$$

$$x = 500,\, y = 1000$$

最优目标函数值 $z = 5,000,000$。此时仅有约束条件 $t_3 \le 500$ 的松弛变量非 0。

整数规划 LINGO 使用混合整数二次规划 MIQP 模型求解，求解器迭代次数 4028，运行时间 0.39s，求解结果如下

$$a_1 = b_1 = 0,\, a_2 = 1500,\, b_2 = 1000,$$

$$t_1 = 500,\, t_2 = 500,\, t_3 = 0,$$

$$x = 500,\, y = 1000$$

最优目标函数值 $z = 5,000,000$。此时仅有约束条件 $t_3 \le 500$ 的松弛变量非 0。

对比上述结果可以发现，两种规划得到了相同的结果，但就求解过程而言，连续规划的迭代次数、运行时间均比整数规划少得多，由此可以看出整数规划算法的效率一般比连续规划要低，这启示我们对于某些问题来说，可以使用连续规划进行试探，如果能找到恰当解，即可采用，从而减少了整数规划带来的额外运算量。

另外，可以发现上述大部分约束条件的松弛变量均为 0，意味着最优点处它们均取等号，这意味着生产时将原油 A，B 的库存均全部消耗，且生产出的产品甲、乙刚好达到合格的最低线（含原油 A 的比例为 50%, 60%）。最后的方案中只生产乙产品而不生产甲。这一现象可以通过定性分析来解释。首先，原油 B 的库存量相比原油 A 更多，生产时应尽可能使用，因为它们不会带来额外的采购成本，而原油 A 应尽量减少额外购买；其次，虽然同等重量的乙产品需要的 A 原油比甲略多，但其售价也比甲要更高，在题给条件下（如原油库存量等），乙的投入产出比更高。因此，使用所有库存，只生产乙产品，且只需让 A 原油含量刚好达到合格线即可，这是比较自然的结论。

3.5 结论

该公司应购买 1000 吨原油 A，并使用完所有的 500 吨 A、1000 吨 B 的库存，全部用来生产汽油乙，此时最大利润为 5 百万元。

4 Ch10-P11 钢管下料

4.1 问题分析与模型建立

根据题意，最多只能有四种切割模式，可以假设第 i 种模式切割得到的 290mm, 315mm, 350mm 以及 455mm 长的钢管分别有 $r_{1i}, r_{2i}, r_{3i}, r_{4i}$ 根。题目中不同切割模式会根据采用次数不同进行收费，因此不妨设四种切割模式使用次数为 x_1, x_2, x_3, x_4，且满足 $x_1 \ge x_2 \ge x_3 \ge x_4$。根据题给条件，可以转化为如下的规划模型（由于没有提供钢管单价，默认假设单价为 1 个单位）：

目标函数

$$\min z = 1.1x_1 + 1.2x_2 + 1.3x_3 + 1.4x_4$$

约束条件如下

客户需求

$$\mathbf{Rx} \ge \begin{bmatrix} 15 \\ 28 \\ 21 \\ 30 \end{bmatrix}$$

其中 $\mathbf{R} = (r_{ij})_{(n\times n)}$，$\mathbf{x} = [x_1, x_2, x_3, x_4]^{\mathrm{T}}$

切割模式的限制

$$\sum_{i=1}^{4} r_{ij} \leq 5,\ \forall j \in \{1,\,2,\,3,\,4\}$$

余料的限制　这里题中要求“每种切割模式下的余料浪费不能超过 100mm”，存在歧义，有下面两种理解方式

1. 每种切割模式切割一根钢管的余料浪费不能超过 100mm；
2. 每种切割模式切割完所有钢管后，总余料不能超过 100mm。

对于理解 1，约束条件为

$$\mathbf{1750} \leq \mathbf{R}^{\mathrm{T}} \begin{bmatrix} 290 \\ 315 \\ 350 \\ 455 \end{bmatrix} \leq \mathbf{1850}$$

加粗的数字代表它们符合不等式的相应维度的列向量。

对于理解 2，约束条件为

$$\mathbf{R}^{\mathrm{T}} \begin{bmatrix} 290 \\ 315 \\ 350 \\ 455 \end{bmatrix} \leq \mathbf{1850}$$

以及总余料约束

$$\left(\mathbf{1850} - \mathbf{R}^{\mathrm{T}} \begin{bmatrix} 290 \\ 315 \\ 350 \\ 455 \end{bmatrix} \right) \mathbf{x} \leq \mathbf{100}$$

上述所有决策变量 r_{ij} 或 x_i 均为非负整数。

4.2 算法设计

上述两种理解得到的模型都可以用 LINGO 的整数规划功能求解，两种理解均为非线性规划。为了加快求解速度，可以考虑增加如下约束

- 采用钢管数量的下界为 $\lceil \frac{15\times290+28\times315+21\times350+30\times455}{1850} \rceil = 19$，因此有

$$\sum_{i=1}^{4} x_i \geq 19$$

- 考虑四种比较简单的方案：1. 2 根 455mm，3 根 290mm；2. 5 根 350mm；3. 4 根 455mm；4. 2 根 315mm，2 根 350mm，1 根 455mm。容易验证，只需要方案 1 使用 8 根钢管，方案 4 使用 14 根钢管即可满足题意，从而得到钢管使用数量的上界 22，有

$$\sum_{i=1}^{4} x_i \leq 22$$

这里的上界约束是只考虑了理解 1，而理解 1 对于余料的约束是没有理解 2 严格的，最优解存在时，理解 2 使用的钢管数目不应多于理解 1，从而理解 1 的上界对理解 2 也是有效的。

计算时可以比较上述约束加入前后对于求解速度以及结果的影响。

4.3 LINGO 代码

对于理解 1，代码如下

```
MODEL:
TITLE: Pipe Cut;
SETS:
  n/1..4/: req, cost, x, len;
  link(n, n): r;
ENDSETS
DATA:
  cost = 1.1 1.2 1.3 1.4;
  req = 15 28 21 30;
  len = 290 315 350 455;
ENDDATA
min = @sum(n : cost * x);
@for( n(i):
  @sum(n(j): x(j) * r(i, j)) >= req(i);
  @sum(n(j): r(j, i)) <= 5;
  @sum(n(j): len(j) * r(j, i)) >= 1750;
  @sum(n(j): len(j) * r(j, i)) <= 1850;
  @gin(x(i)); @for(n(j): @gin(r(i, j)));
);
! Try to enable/disable the following constraints;
!@sum(n(i): x(i)) >= 19;
!@sum(n(i): x(i)) <= 22;
END
```

求解时，可以选择是否注释掉最后两个约束，以测试不同情况下的求解速度。

对于理解 2，只需要将余料约束按如下修改即可

```
  @sum(n(j): len(j) * r(j, i)) <= 1850;
  x(i) * (1850 - @sum(n(j): len(j) * r(j, i))) <= 100;
```

4.4 计算结果与分析

对于上述两种理解，LINGO 均是使用 PIQP，即纯整数二次规划模型，求解器种类为 B-and-B 即分支定界 (`Global Solver` 关闭时)，否则为 Global。

对于理解 2，如果加上额外的上界/下界约束，`Global Solver` 求解器迭代 277005 次后得不到可行解 (Infeasible)；如果不加上该约束，`Global Solver` 求解器迭代 100 万次仍不能终止，同时也找不到可行解。两种情况表明这种理解应该是不符合出题者本意的。

对于理解 1，求解时迭代次数如下

	额外约束存在	额外约束不存在
`Global Solver` On	41909	2180406*
`Global Solver` Off	2600	14045

表 1: 不同设定下求解器迭代次数

上述 * 符号代表该种情况已经迭代了相应次数，但计算还未终止。

3 种在有限时间内收敛的情况的目标函数最优值均为 $z = 21.5$，即单根钢管价格为 1 个单位时，最小费用为 21.5，且得到的四种方案采用的钢管数相同，分别为

$$x_1 = 14, x_2 = 4, x_3 = 1, x_4 = 0$$

在使用 Local Solver 的情况中，虽然软件报告的是 `Local optimal solution found.`，但得到的局部最优解的目标函数值与 Global Solver 得到的结果相同，说明它也是全局最优解。

可以发现，最优解一共使用了 19 根钢管，这恰好是上面提到的额外约束条件中的所需钢管数目的下界。

不过，最优解对应的切割方式有所不同，如下

序号	**290mm**	**315mm**	**350mm**	**455mm**
1	1	2	0	2
2	0	0	5	0
3	2	0	1	2
4	2	0	1	2

表 2: 最优切割方式 (Local Solver, 不带附加约束)

序号	**290mm**	**315mm**	**350mm**	**455mm**
1	1	2	0	2
2	0	0	5	0
3	2	0	1	2
4	2	1	0	2

表 3: 最优切割方式 (Local Solver, 带附加约束)

序号	**290mm**	**315mm**	**350mm**	**455mm**
1	1	2	0	2
2	0	0	5	0
3	2	0	1	2
4	0	0	0	4

表 4: 最优切割方式 (Global Solver, 带附加约束)

但考虑到实际上第四种方案并未被使用，因此实际上这三种解是等价的，因为它们提供的前三种方案均完全相同。

同时，比较上面不同设置时的求解器迭代次数，可以发现添加总钢管数量的上/下界约束后，无论是 Local Solver 还是 Global Solver，迭代次数均有了显著下降。实际上，约束 $19 \leq \sum_{i=1}^{4} x_i \leq 22, x_i \in \mathbb{Z}$ 的解空间的大小可以精确计算，容易得到它共有 $\sum_{i=22}^{25} \binom{i}{3} = 7635$ 组解，较为有限，从而可以加快迭代速度。而不使用该约束时，Global Solver 迭代 200 万次以上均未得到全局最优解，这说明在求解整数规划时，可以通过考虑极端情况，手动计算以增加合适的边界条件，从而缩小解空间范围，使得程序更易求解。

本题中的 Local Solver 迭代次数远比 Global Solver 少，这种现象在通常情况下是存在的。而局部最优解就是全局最优解并不是在所有情况下均能成立的。最佳的方法仍应该是保留上界/下界约束，使用 Global Solver 进行求解，保证得到全局最优解。

4.5 结论

题中要求的余料浪费不超过 100mm 的约束应针对每个切割模式切割一根钢管的情况，否则问题无解。有解的情况下，最小的总费用是钢管单价的 21.5 倍，最优切割方案为：

1. 1 根 290mm，2 根 315mm，2 根 455mm；
2. 5 根 350mm；
3. 2 根 290mm，1 根 350mm，2 根 455mm。

使用三种方案分别购买并切割 14 根、4 根、1 根钢管即可满足要求。

5 收获与建议

通过本实验，我对 LINGO 软件的使用更加熟练，对整数规划有了更深刻的理解。建议：更新本次实验第 11 题的题面描述，避免出现二义性。